WEEK 15: TAYLOR AND FOURIER SERIES

In mathematics, it is often helpful to approximate complicated functions by simpler ones. In this lecture, we consider two important illustrations of this fundamental idea. First, we examine the approximation of functions by *polynomials* via Taylor's Theorem. This kind of approximation is useful when one is interested in the behaviour of the complicated function in a small neighbourhood of a particular point. Second, we use *trigonometric polynomials* to approximate functions that are known to be *periodic*. This leads to the Fourier series studied in Calculus.

1. Taylor's Theorem

Theorem 1.1. *Let $f : [a,b] \to \mathbb{R}$ be continuous on $[a,x] \subset [a,b]$ and N times differentiable on (a,x). Then, there exists $\xi \in (a,x)$ such that*

$$f(x) = \underbrace{\sum_{n=0}^{N-1} \frac{f^{(n)}(a)}{n!}(x-a)^n}_{T_N(x)} + \underbrace{\frac{f^{(N)}(\xi)}{N!}(x-a)^N}_{R_N(x)} .$$

Note the following:

- a is called the *point of expansion.*
- T_N is called the *Taylor polynomial of degree $< N$ of f* at a.
- In general, ξ will depend on x.
- R_N is called the *remainder.*
- The coefficient

$$c_n := \frac{f^{(n)}(a)}{n!} \tag{1.1}$$

  of the Taylor polynomial is called the *nth Taylor coefficient of f* at a.
- We have

$$\lim_{x \to a} R_N(x) = 0 .$$

  Hence, in the neighbourhood of a, the Taylor polynomial T_N is a good approximation of f.

Here's a first attempt to write a procedure for calculating the Taylor polynomial T_N of f at the point of expansion a:

```
taylorPolynomial := proc(f,a,N)
    local n,x,c_n,T :
    T := eval(f(a)) :
    for n from 1 to N-1 do
        c_n := eval(subs(x=a,diff(f(x),x$n))/n!) :
        T := T + c_n*x^n :
    od :
    return unapply(T,x) :
end proc :
```

```
f := x -> cos(x);
                                    f := cos

> T_5 := taylorPolynomial(f,0,5);

                                             2          4
                    T_5 := x -> 1 - 1/2 x  + 1/24 x

> plot({T_5(x),f(x)},x=-Pi..Pi);
```



FIGURE 1. The cosine function and its Taylor polynomial T_5 when $a = 0$.

2. Calculation of the Taylor coefficients

The procedure works, but is needlessly wasteful; for each value of n, the nth derivative of f is calculated "from scratch", as are $n!$ and $(x-a)^n$. A better algorithm should re-use the work already done by exploiting the fact that

$$f^{(n)}(x) = \frac{\mathrm{d}}{\mathrm{d}x} f^{(n-1)}(x) \quad \text{and} \quad \frac{(x-a)^n}{n!} = \frac{x-a}{n} \frac{(x-a)^{n-1}}{(n-1)!}\,.$$

```
taylorPolynomial := proc(f,a,N)
     local n,x,factor,F,T:
     factor := 1 :
     F := f(x) :
     T := eval(subs(x=a,F)) :
     for n from 1 to N-1 do
         F := diff(F,x) :
         factor := factor*(x-a)/n :
```

```
        T := T + eval(subs(x=a,F))*factor:
    od :
    return unapply(T,x) :
end proc :
```

Furthermore, in many practical situations, it is not even necessary to differentiate f repeatedly in order to calculate the Taylor coefficients. We now elaborate this point.

First, we note that there is no loss of generality in taking $a = 0$ (we can always shift the origin by a simple change of the independent variable). Let us assume that f is infinitely differentiable at 0 and that

$$\lim_{N\to\infty} R_N(x) = 0\,. \tag{2.1}$$

Then

$$f(x) = \sum_{n=0}^{\infty} c_n x^n\,. \tag{2.2}$$

The series on the right-hand side is called the *Taylor series of f* (at the origin).

Example 2.1. *Find the first N Taylor coefficients of the function $y = f(x) = \cos x$. We write*

$$y = \sum_{n=0}^{\infty} c_n x^n$$

and use the fact that

$$\frac{d^2y}{dx^2} = -y, \quad y(0) = 1, \ \ y'(0) = 0\,. \tag{2.3}$$

On the left-hand side of the differential equation, we have

$$\frac{d^2y}{dx^2} = \sum_{n=0}^{\infty} n(n-1)c_n x^{n-2} = \sum_{n=0}^{\infty}(n+2)(n+1)c_{n+2}x^n\,.$$

On the right-hand side, we have

$$-y = \sum_{n=0}^{\infty} -c_n x^n\,.$$

Equating the two series term by term gives

$$(n+2)(n+1)c_{n+2} = -c_n, \quad n \geq 0\,. \tag{2.4}$$

The initial conditions $y(0) = 1$ and $y'(0) = 0$ give $c_0 = 1$ and $c_1 = 0$. The recurrence relation (2.4) then enables us to obtain c_2, c_3 etc.

3. Fourier series

We say that a function $f : \mathbb{R} \to \mathbb{R}$ is periodic if there exists a number $p > 0$ such that

$$\forall\, x \in \mathbb{R}, \quad f(x+p) = f(x)\,. \tag{3.1}$$

The *period* of f is then the smallest positive number p for which this holds.

You are already familiar with many periodic functions such as $\sin x$ and $\cos x$ (of period 2π). Periodic functions occur naturally in real life. For instance, the position of the Earth relative to the Sun is a periodic function of time; the period p is approximately a calendar year.

We can always rescale the independent variable to make the period of a periodic function exactly equal to 2π. Henceforth, we shall therefore assume that

$$p = 2\pi\,.$$

Given a periodic function f, we define its *Fourier series* by

$$\frac{a_0}{2} + \sum_{n=0}^{\infty} \{a_n \cos(nx) + b_n \sin(nx)\}\,, \tag{3.2}$$

where

$$a_n = \frac{1}{\pi}\int_{-\pi}^{\pi} f(x)\cos(nx)\,\mathrm{d}x, \quad b_n = \frac{1}{\pi}\int_{-\pi}^{\pi} f(x)\sin(nx)\,\mathrm{d}x\,. \tag{3.3}$$

As discussed in Calculus, a rough statement of Fourier's theorem for periodic functions is as follows:

Theorem 3.1. *Suppose that f is piecewise smooth, then*

$$\frac{f(x+)+f(x-)}{2} = \frac{a_0}{2} + \sum_{n=0}^{\infty} \{a_n \cos(nx) + b_n \sin(nx)\}\,,$$

where

$$f(x\pm) = \lim_{0<\varepsilon\to 0} f(x\pm\varepsilon)\,.$$

The trigonometric polynomial

$$F_N = \frac{a_0}{2} + \sum_{n=0}^{N-1} \{a_n \cos(nx) + b_n \sin(nx)\}$$

therefore provides an approximation of f.

```
fourierPolynomial := proc(f,N)
    local n,a,b,x,F :
    a := eval(int(f(x),x=-Pi..Pi)/Pi) :
    F := a/2 :
    for n from 1 to N-1 do
        a := eval(int(f(x)*cos(n*x),x=-Pi..Pi)/Pi) :
        b := eval(int(f(x)*sin(n*x),x=-Pi..Pi)/Pi) :
        F := F + a*cos(n*x) + b*sin(n*x) :
    od :
    return unapply(F,x) :
end proc :
```



FIGURE 2. The absolute value function $|x|$ and its Fourier trigonometric polynomial F_4.